******************************************************************************************************
***Analysis 1 (Overall Sample)

****** Effects of Information Provision (without considering misperception): Panel A of Figure 3

use analysis.dta, clear
   
reg Q19_concernforyou treatment NEP
estimates store concern_self                      

reg Q20_concernfornation treatment NEP
estimates store concern_nation 

reg Q23_1_normgov treatment NEP
estimates store norm_gov  
 
reg Q23_2_normperson treatment NEP                          
estimates store norm_person 
 
reg consumer_behav treatment NEP
estimates store consumer 

reg willing_pay treatment NEP
estimates store willingness                      

reg asso_resp treatment NEP
estimates store group                         

reg poli_behav treatment NEP
estimates store political                           

coefplot  ///
   (concern_self, rename(treatment = "Concern for self and family")) ///
   (concern_nation, rename(treatment = "Concern for the country")) ///
   (norm_gov, rename(treatment = "Government should do more")) ///
   (norm_person, rename(treatment = "Personal obligation")) ///
   (consumer, rename(treatment = "Consumer behaviors")) ///
   (willingness, rename(treatment = "Policy support/Willingness to pay")) ///
   (group, rename(treatment = "Group participation and contribution")) ///
   (political, rename(treatment = "Political activities")) ///
   ,keep(treatment) offset(0) legend(off) levels(95 90) ciopts(color(black black) lcolor(*.5 *.5)) ///
    msymbol(D) mcolor(black) msize(small) ///
    xline(0, lwidth(thin) lpattern(dash) lcolor(black)) ylabel(,labsize(small)) xtitle("Marginal Effect of Treatment", size(small)) ///
    grid(none) coeflabels(,notick labgap(2)) ///
    headings("Concern for self and family" = "{bf:Issue-specific Environmental Concern}" "Government should do more" = "{bf:Norms/Attitudes}" "Consumer behaviors" = "{bf:Preferences and Behavioral Intentions}", nogap labgap(2)) ///
    graphregion(fcolor(white))


****** Effects of Information Provision by Misperception: Panels B - D of Figure 3; Table 1

use analysis.dta, clear
keep if misbelief > -80 & misbelief < 40
 
reg Q19_concernforyou i.treatment c.misbelief i.treatment#c.misbelief NEP
estimates store concern_self_reg
margins, dydx(treatment) at(misbelief=(-80(5)40)) post
estimates store concern_self

reg Q20_concernfornation i.treatment c.misbelief i.treatment#c.misbelief NEP 
estimates store concern_nation_reg
margins, dydx(treatment) at(misbelief=(-80(5)40)) post
estimates store concern_nation


reg Q23_1_normgov i.treatment c.misbelief i.treatment#c.misbelief NEP 
estimates store norm_gov_reg
margins, dydx(treatment) at(misbelief=(-80(5)40)) post
estimates store norm_gov

reg Q23_2_normperson i.treatment c.misbelief i.treatment#c.misbelief NEP 
estimates store norm_person_reg
margins, dydx(treatment) at(misbelief=(-80(5)40)) post
estimates store norm_person

reg consumer_behav i.treatment misbelief i.treatment#c.misbelief NEP 
estimates store consumer_reg  
margins, dydx(i.treatment) at(misbelief=(-80(5)40)) post
estimates store consumer    
                           
reg willing_pay i.treatment misbelief i.treatment#c.misbelief NEP
estimates store willingness_reg 
margins, dydx(i.treatment) at(misbelief=(-80(5)40)) post
estimates store willingness 
                            
reg asso_resp i.treatment misbelief i.treatment#c.misbelief NEP 
estimates store group_reg  
margins, dydx(i.treatment) at(misbelief=(-80(5)40)) post
estimates store group    
                          
reg poli_behav i.treatment misbelief i.treatment#c.misbelief NEP 
estimates store political_reg 
margins, dydx(i.treatment) at(misbelief=(-80(5)40)) post
estimates store political 

#delimit ;   
coefplot
    (concern_self), bylabel(Concern for Self and Family) || (concern_nation), bylabel(Concern for the Country) ||, 
     at byopts(title("Issue-specific Environmental Concern", size(medium)))

     ytitle("Marginal Effect of Treatment", size(small)) 
     xtitle("Misperception (Perceived Ranking of Risk - Actual Ranking of Risk)", size(small))
     recast(line) lcolor(black) lwidth(*1.5) ciopts(recast(rarea) color(gray) color(%20) lcolor(%0))
     yline(0, lwidth(thin) lpattern(dash) lcolor(black))
     graphregion(fcolor(white))
;
#delimit cr


#delimit ;   
coefplot
    (norm_gov), bylabel(Government Should Do More) || (norm_person), bylabel(Personal Obligation) ||, 
     at byopts(title("Norms/Attitudes", size(medium)))

        ytitle("Marginal Effect of Treatment", size(small)) 
        xtitle("Misperception (Perceived Ranking of Risk - Actual Ranking of Risk)", size(small))
         recast(line) lcolor(black) lwidth(*1.5) ciopts(recast(rarea) color(gray) color(%20) lcolor(%0))
	 yline(0, lwidth(thin) lpattern(dash) lcolor(black))
	 graphregion(fcolor(white))
;
#delimit cr


#delimit ;   
coefplot
    (consumer), bylabel(Consumer Behaviors) || (willingness), bylabel(Policy Support/Willingness to Pay) || 
    (group), bylabel(Group Participation and Contribution) || (political), bylabel(Political Activities) ||, 
     at byopts(title("Preferences and Behavioral Intentions", size(medium)))
     
        ytitle("Marginal Effect of Treatment", size(small)) 
        xtitle("Misperception (Perceived Ranking of Risk - Actual Ranking of Risk)", size(small))
         recast(line) lcolor(black) lwidth(*1.5) ciopts(recast(rarea) color(gray) color(%20) lcolor(%0))
	 yline(0, lwidth(thin) lpattern(dash) lcolor(black))
	 graphregion(fcolor(white))
;
#delimit cr


esttab concern_self_reg concern_nation_reg norm_gov_reg norm_person_reg consumer_reg willingness_reg group_reg political_reg ///
       using treatmentbyMisperception_overall.rtf, ///
       b(3) se(3) replace compress r2(3) star(* 0.10 ** 0.05 *** 0.01) nodepvar